/* 以太网控制舵机
 * 目的:该例程为以太网控制舵机的例程
 * 作者:Zeeland
 * 最后修改时间:2021年11月9日 18:33:34
 * https://gitee.com/zeeland/projects
 */
 
#include <Servo.h>
#include <ESP8266WiFi.h>        // 本程序使用 ESP8266WiFi库
#include <ESP8266WiFiMulti.h>   //  ESP8266WiFiMulti库
#include <ESP8266WebServer.h>   //  ESP8266WebServer库

Servo myservo;                  // 实例化一个舵机
ESP8266WiFiMulti wifiMulti;     // 建立ESP8266WiFiMulti对象,对象名称是 'wifiMulti'
//保存html页面
String htmlStr = "<form action=\"/OpenServo\" method=\"POST\"><input type=\"submit\" value=\"Open Servo\"></form>";

ESP8266WebServer esp8266_server(80);// 建立网络服务器对象，该对象用于响应HTTP请求。监听端口（80）

//旋转特定角度，传入angle为需要旋转的角度，dir为方向
void rotateSpecificAngle(int angle,int dir){
  int pos;
  if(dir){
    for(pos = 0; pos < angle; pos += 3)  
    {         
       // 每次步进一度   
       // 指定舵机转向的角度                                        
      myservo.write(pos);
      // 等待15ms让舵机到达指定位置        
      delay(15);                       
    }  
  }else{
    for(pos = angle; pos >= 1; pos -= 3)   //从180度到0度运动  
    {                                
      myservo.write(pos);         // 指定舵机转向的角度
      delay(15);                        // 等待15ms让舵机到达指定位置
    }
  }
}

// 开始舵机工作
void servoWork() {
  rotateSpecificAngle(180,0);
  rotateSpecificAngle(180,1);
}


/*设置服务器根目录即'/'的函数'handleRoot'
  该函数的作用是每当有客户端访问NodeMCU服务器根目录时，
  NodeMCU都会向访问设备发送 HTTP 状态 200 (Ok) 这是send函数的第一个参数。
  同时NodeMCU还会向浏览器发送HTML代码，以下示例中send函数中第三个参数，
  也就是双引号中的内容就是NodeMCU发送的HTML代码。该代码可在网页中产生LED控制按钮。 
  当用户按下按钮时，浏览器将会向NodeMCU的/LED页面发送HTTP请求，请求方式为POST。
  NodeMCU接收到此请求后将会执行handleLED函数内容*/
void handleRoot() {       
  Serial.println("root response send");
  esp8266_server.send(200, "text/html", htmlStr);
}

//处理Servo控制请求的函数'handleOpenServo'
void handleOpenServo() {                          
  Serial.println("OpenServo POST response send");

  servoWork();                                        // 舵机开始工作
  esp8266_server.sendHeader("Location","/");          // 跳转回页面根目录
  esp8266_server.send(303);                           // 发送Http相应代码303 跳转  
}
 
// 设置处理404情况的函数'handleNotFound'
void handleNotFound(){
  Serial.println("404 response send");
  esp8266_server.send(404, "text/plain", "404: Not found"); // 发送 HTTP 状态 404 (未找到页面) 并向浏览器发送文字 "404: Not found"
}


void setup() {
  
  myservo.attach(2);  // 舵机绑定GPIO2引脚，GPIO2对应D4引脚
  
  Serial.begin(9600);   // 启动串口通讯
  
  wifiMulti.addAP("ssid_from_AP_1", "your_password_for_AP_1"); // 将需要连接的一系列WiFi ID和密码输入这里
  wifiMulti.addAP("ssid_from_AP_2", "your_password_for_AP_2"); // ESP8266-NodeMCU再启动后会扫描当前网络
  wifiMulti.addAP("ssid_from_AP_3", "your_password_for_AP_3"); // 环境查找是否有这里列出的WiFi ID。如果有
  Serial.println("Connecting ...");                            // 则尝试使用此处存储的密码进行连接。
  
  int i = 0;                                 
  while (wifiMulti.run() != WL_CONNECTED) {  // 此处的wifiMulti.run()是重点。通过wifiMulti.run()，NodeMCU将会在当前
    delay(1000);                             // 环境中搜索addAP函数所存储的WiFi。如果搜到多个存储的WiFi那么NodeMCU
    Serial.print(i++); Serial.print(' ');    // 将会连接信号最强的那一个WiFi信号。
  }                                          // 一旦连接WiFI成功，wifiMulti.run()将会返回“WL_CONNECTED”。这也是
                                             // 此处while循环判断是否跳出循环的条件。

  // WiFi连接成功后将通过串口监视器输出连接成功信息 
  Serial.println('\n');
  Serial.print("Connected to ");
  Serial.println(WiFi.SSID());              // 通过串口监视器输出连接的WiFi名称
  Serial.print("IP address:\t");
  Serial.println(WiFi.localIP());           // 通过串口监视器输出ESP8266-NodeMCU的IP

  esp8266_server.begin();                           // 启动网站服务
  esp8266_server.on("/", HTTP_GET, handleRoot);     // 设置服务器根目录即'/'的函数'handleRoot'
  esp8266_server.on("/OpenServo", HTTP_POST, handleOpenServo);  // 设置处理LED控制请求的函数'OpenServo'
  esp8266_server.onNotFound(handleNotFound);        // 设置处理404情况的函数'handleNotFound'

  Serial.println("HTTP esp8266_server started");//  告知用户ESP8266网络服务功能已经启动

}

void loop() {
  esp8266_server.handleClient();                     // 检查http服务器访问
}
